Computergestuetzte Mathematik zur Analysis

Lektion 8 (10. Dez.)

```
> restart:
```

Einige Integrale

```
> a:= Int((1-t^4)^(-1/2),t=0..1); # Betafunktion
```

$$a := \int_0^1 \frac{1}{\sqrt{-t^4+1}}\, dt \quad (1.1)$$

```
> value(a);
```

$$\frac{1}{4} \mathrm{B}\left(\frac{1}{4}, \frac{1}{2}\right) \quad (1.2)$$

```
> b:= Int(t^2*(1-t^2)^(-1/2),t=0..1);
```

$$b := \int_0^1 \frac{t^2}{\sqrt{-t^2+1}}\, dt \quad (1.3)$$

```
> value(b);
```

$$\frac{1}{4} \pi \quad (1.4)$$

```
> A := Int(exp(-x^2), x = -infinity .. infinity);
```

$$A := \int_{-\infty}^{\infty} e^{-x^2}\, dx \quad (1.5)$$

A) 1

B) $\sqrt{\pi}$

```
> value(A);
```

$$\sqrt{\pi} \quad (1.6)$$

```
> B := Int(exp(-x^2), x);
```

$$B := \int e^{-x^2}\, dx \quad (1.7)$$

```
> value(B);
```

$$\frac{1}{2}\sqrt{\pi}\,\mathrm{erf}(x) \quad (1.8)$$

```
> C := Int(ln(x)^2 / (1 + x)^2, x);
```

$$C := \int \frac{\ln(x)^2}{(1+x)^2}\, dx \quad (1.9)$$

```
> value(C);
```

$$\int \frac{\ln(x)^2}{(1+x)^2}\, dx \quad (1.10)$$

```
> E := Int(ln(x)^2 / (1 + x)^2, x = 0 .. infinity);
```

$$E := \int_0^{\infty} \frac{\ln(x)^2}{(1+x)^2}\, dx \quad (1.11)$$

```
> value(E);
```

$$\frac{1}{3}\pi^2 \quad (1.12)$$

A := evalf(E, 15);

B := evalf(value(E),15):

1) A und B stimme uebeerein

2) A ist genauer als B

3) B ist genauer als A

```
> evalf(E, 15);
```

3.28986813369652 **(1.13)**

```
> evalf(value(E), 15);
```

3.28986813369644 **(1.14)**

```
> evalf(value(E),20);
  evalf(E,20);
```

3.2898681336964528730

3.2898681336964528729 **(1.15)**

Riemann Integral

```
> restart:
> S := Sum(a/n*exp(k*a/n), k = 0 .. n-1);
```

$$S := \sum_{k=0}^{n-1} \frac{a\,e^{\frac{k\,a}{n}}}{n} \tag{2.1}$$

```
> value(S);
```

$$\frac{a\,e^{a}}{\left(e^{\frac{a}{n}}-1\right)n} - \frac{a}{\left(e^{\frac{a}{n}}-1\right)n} \tag{2.2}$$

```
> L := Limit(S, n = infinity);
```

$$L := \lim_{n\to\infty} \left(\sum_{k=0}^{n-1} \frac{a\,e^{\frac{k\,a}{n}}}{n}\right) \tag{2.3}$$

```
> value(L);
```

$$e^{a}-1 \tag{2.4}$$

Grenzwerte

```
> restart:
> a := k^2/(2^k);
```

$$a := \frac{k^2}{2^k} \tag{3.1}$$

1) Die Folge der a_k konvergiert gegen 0

2) Die Folger der a_k divergiert

3) Die Folge der a_k konvergiert gegen $\sqrt{\pi}$

```
> A := Limit(a, k = infinity);
```

$$A := \lim_{k\to\infty} \frac{k^2}{2^k} \tag{3.2}$$

```
> value(A);
```

$$0 \tag{3.3}$$

```
> b := k^k/k!;
```

$$b := \frac{k^k}{k!} \tag{3.4}$$

1) Die Folge der b_k konvergiert gegen 0

2) Die Folge der b_k divergiert

3) Die Folge der b_k konvergiert gegen $\sqrt{2}$

```
> B := Limit(b, k = infinity);
```

$$B := \lim_{k \to \infty} \frac{k^k}{k!} \quad (3.5)$$

```
> value(B);
```

$$\infty \quad (3.6)$$

```
> c := (-1)^k*b;
```

$$c := \frac{(-1)^k k^k}{k!} \quad (3.7)$$

```
> C := Limit(c, k = infinity);
```

$$C := \lim_{k \to \infty} \frac{(-1)^k k^k}{k!} \quad (3.8)$$

```
> value(C);
```

undefined (3.9)

```
> e := sin(k*Pi);
```

$$e := \sin(k\pi) \quad (3.10)$$

```
> E := Limit(e, k = infinity);
```

$$E := \lim_{k \to \infty} \sin(k\pi) \quad (3.11)$$

```
> value(E);
```

$$-1..1 \quad (3.12)$$

```
> value(E) assuming k::integer;
```

$$-1..1 \quad (3.13)$$

```
> e assuming k::integer;
```

$$0 \quad (3.14)$$

Reihen

```
> restart:
> a := 1/k/(k+1);
```

$$a := \frac{1}{k(k+1)} \quad (4.1)$$

```
> A := Sum(a, k = 1 .. infinity);
```

$$A := \sum_{k=1}^{\infty} \frac{1}{k(k+1)} \quad (4.2)$$

```
> value(A);
```

$$1 \quad (4.3)$$

```
> A1 := Sum(a, k = 1 .. N);
```

$$A1 := \sum_{k=1}^{N} \frac{1}{k\,(k+1)} \tag{4.4}$$

```
> value(A1);
```

$$-\frac{1}{N+1} + 1 \tag{4.5}$$

```
> b := 1/k^3;
```

$$b := \frac{1}{k^3} \tag{4.6}$$

```
> B := Sum(b, k = 1 .. infinity);
```

$$B := \sum_{k=1}^{\infty} \frac{1}{k^3} \tag{4.7}$$

```
> value(B);
```

$$\zeta(3) \tag{4.8}$$

```
> evalf(%);
```

$$1.202056903 \tag{4.9}$$

Produkte

```
> p:= Product(1-1/k^2,k=2..infinity);
```

$$p := \prod_{k=2}^{\infty} \left(1 - \frac{1}{k^2}\right) \tag{5.1}$$

```
> value(p);
```

$$\frac{1}{2} \tag{5.2}$$

```
> p:= Product(1-1/(2*k)^2,k=1..infinity); # Wallis Formel
```

$$p := \prod_{k=1}^{\infty} \left(1 - \frac{1}{4\,k^2}\right) \tag{5.3}$$

```
> value(p);
```

$$\frac{2}{\pi} \tag{5.4}$$

```
> product(GAMMA(k/3),k=1..8);
```

$$\frac{640}{6561}\,\pi^3\sqrt{3} \tag{5.5}$$

Gleichmaessige Konvergenz

```
> a := (4*sin(x)*(1/5))^k;
```

$$a := \left(\frac{4}{5}\sin(x)\right)^k \tag{6.1}$$

```
> S := Sum(a, k = 1 .. infinity);
```

$$S := \sum_{k=1}^{\infty} \left(\frac{4}{5}\sin(x)\right)^k \tag{6.2}$$

```
> f := value(S);
```

$$f := -\frac{4\sin(x)}{4\sin(x) - 5} \tag{6.3}$$

```
> farben := [black, red, yellow, blue, green, magenta, coral,
  pink, cyan];
```

$$farben := [black, red, yellow, blue, green, magenta, coral, pink, cyan] \tag{6.4}$$

```
> funktionen := [f, seq(sum(a, k = 1 .. n), n = 1 .. 7)];
```

$$funktionen := \left[-\frac{4\sin(x)}{4\sin(x)-5}, \frac{4}{5}\sin(x), \frac{4}{5}\sin(x) + \frac{16}{25}\sin(x)^2, \frac{4}{5}\sin(x) + \frac{16}{25}\sin(x)^2 + \frac{64}{125}\sin(x)^3, \frac{4}{5}\sin(x) + \frac{16}{25}\sin(x)^2 + \frac{64}{125}\sin(x)^3 + \frac{256}{625}\sin(x)^4, \frac{4}{5}\sin(x) + \frac{16}{25}\sin(x)^2 + \frac{64}{125}\sin(x)^3 + \frac{256}{625}\sin(x)^4 + \frac{1024}{3125}\sin(x)^5, \frac{4}{5}\sin(x) + \frac{16}{25}\sin(x)^2 + \frac{64}{125}\sin(x)^3 + \frac{256}{625}\sin(x)^4 + \frac{1024}{3125}\sin(x)^5 + \frac{4096}{15625}\sin(x)^6, \frac{4}{5}\sin(x) + \frac{16}{25}\sin(x)^2 + \frac{64}{125}\sin(x)^3 + \frac{256}{625}\sin(x)^4 + \frac{1024}{3125}\sin(x)^5 + \frac{4096}{15625}\sin(x)^6 + \frac{16384}{78125}\sin(x)^7\right] \tag{6.5}$$

```
> plot(funktionen, x = -Pi .. Pi, color = farben);
```




```
> funktionen := [f, f+1/3, f-1/3, seq(sum(a, k = 1 .. 5*n), n = 1
  .. 4)]:
> farben := [black, gray, gray, red, yellow, blue, green,
  magenta, coral, pink, cyan];
```

$farben := [black, gray, gray, red, yellow, blue, green, magenta, coral, pink, cyan]$ **(6.6)**

```
> plot(funktionen, x = -Pi .. Pi, color = farben, thickness = 2,
  numpoints = 500);
```



Das Taylorpolynom

```
> f := sqrt(1+x)/sqrt(1+x^2);
```

$$f := \frac{\sqrt{1+x}}{\sqrt{x^2+1}} \quad \textbf{(7.1)}$$

```
> t := series(f, x  = 0, 8);
```

$$t := 1 + \frac{1}{2}\,x - \frac{5}{8}\,x^2 - \frac{3}{16}\,x^3 + \frac{51}{128}\,x^4 + \frac{47}{256}\,x^5 - \frac{369}{1024}\,x^6 - \frac{267}{2048}\,x^7 + \mathrm{O}(x^8) \quad \textbf{(7.2)}$$

```
> P := convert(t, polynom);
```

$$P := 1 + \frac{1}{2}\,x - \frac{5}{8}\,x^2 - \frac{3}{16}\,x^3 + \frac{51}{128}\,x^4 + \frac{47}{256}\,x^5 - \frac{369}{1024}\,x^6 - \frac{267}{2048}\,x^7 \quad \textbf{(7.3)}$$

```
> for n from 1 to 3 do;
>   t := series(f, x = 0, n + 1);
>   P[n] := convert(t, polynom);
> od;
```

$$t := 1 + \frac{1}{2}x + O(x^2)$$

$$P_1 := 1 + \frac{1}{2}x$$

$$t := 1 + \frac{1}{2}x - \frac{5}{8}x^2 + O(x^3)$$

$$P_2 := 1 + \frac{1}{2}x - \frac{5}{8}x^2$$

$$t := 1 + \frac{1}{2}x - \frac{5}{8}x^2 - \frac{3}{16}x^3 + O(x^4)$$

$$P_3 := 1 + \frac{1}{2}x - \frac{5}{8}x^2 - \frac{3}{16}x^3 \qquad (7.4)$$

```
> P[0] := f;
```

$$P_0 := \frac{\sqrt{1+x}}{\sqrt{x^2+1}} \qquad (7.5)$$

```
> farbe := [black,blue, red, cyan];
```

$$farbe := [black, blue, red, cyan] \qquad (7.6)$$

```
> plot(convert(P, list), x = -.8 .. .8, color = farbe, thickness
  = 2);
```




```
> g := cos(x);
```

$$g := \cos(x) \tag{7.7}$$

```
> for n in [4, 20, 60] do;
>   Q[n] := convert(series(g, x, n+1), polynom):
    E[n] := Q[n] - g;
> od:
> Q[4];
```

$$1 - \frac{1}{2}\, x^2 + \frac{1}{24}\, x^4 \tag{7.8}$$

```
> E[0] := 0;
```

$$E_0 := 0 \tag{7.9}$$

```
> Q[0] := g;
```

$$Q_0 := \cos(x) \tag{7.10}$$

```
> plot(convert(Q, list), x = - 32 .. 32, y = -2 .. 2, color =
  farbe, thickness = 2, numpoints = 1000);
```



Das komplexe Bild

```
> restart:
> h := arctan(x);
```

$$h := \arctan(x) \tag{8.1}$$

```
> for n in [4, 20, 60] do;
>   S[n] := convert(series(h, x = 0, n+1), polynom):
> od:
> S[0] := h;
```

$$S_0 := \arctan(x) \tag{8.2}$$

```
> plot(convert(S, list), x = -2 .. 2, -2 .. 2, thickness = 2,
  numpoints = 500);
```




```
> plot([h, S[60]], x = -2 .. 2, -2 .. 2, color = [blue, green],
  thickness = 2, numpoints = 500);
```




```
> x := a + I*b;
```

$$x := a + \mathrm{I}\, b \tag{8.3}$$

```
> pl1 := plot3d(abs(h-S[20]), a = -1.3 .. 1.3, b = -1.3 .. 1.3,
  shading = zhue,  style = patchcontour, numpoints = 2000):
> with(plots):
> display(pl1, axes = boxed, view = 0 .. 5, orientation = [-110,
  35]);
```




```
> pl2 := plot3d(abs(h), a = -1.3 .. 1.3, b = -1.3 .. 1.3, color =
  black,  style = wireframe, numpoints = 6000):
> display([pl1, pl2], axes = boxed, view = 0 .. 2, orientation =
  [-110, 35]);
```




```
> x := 'x':
> plot(abs(arctan(I*x)),x=-2..2);
```

$$\sqrt{2} \quad (8.4)$$